RDP Mathematik und angewandte Mathematik

(%i1) kill(all);

(%o0) *done*

1. Aufgabe (Kosten- und Preistheorie)

Für einen Monopolbetrieb wurde die lineare Kostenfunktion K = 28 x + 900 und die lineare Nachfragefunktion p = 480 - 0,5 x ermittelt.
Zu berechnen sind
a) der Cournotsche Punkt,
b) der maximale Gewinn,
c) die erlösmaximierende Menge und der zugehörige Preis,
d) der maximale Erlös,
e) die Gewinngrenzen,
f) die Sättigungsmenge und
g) die Preisobergrenze.

Wie lauten die Antworten auf diese Fragen, wenn
K = -0,005*x**2+84*x+115200 und p = 520 - 0,13 x ?

GEGEBEN IST

(%i1) K:K(x):=28*x+900;

(%o1) $K(x):=28x+900$

(%i2) p:p(x):=480-0.5*x;

(%o2) $p(x):=480-0.5x$

a) Man berechne den Cournot'schen Punkt.
Der Cournot'sche Punkt hat zwei Koordinaten:
- die Cournot'sche Menge (die gewinnmaximale Menge) und
- den Cournot'schen Preis (man muss die Cournot'sche Menge in die Nachfragefunktion einsetzen.

(%i3) U(x):=p(x)*x,expand;

(%o3) $U(x):=p(x)x$

(%i4) G(x):=U(x)-K(x);

(%o4) $G(x):=U(x)-K(x)$

(%i5) ab:diff(G(x),x);
(%o5) $452 - 1.0\, x$

(%i6) l:realroots(ab);
(%o6) $[\, x = 452\,]$

(%i7) XC:x,l;
(%o7) 452

(%i8) PC:p(XC);
(%o8) 254.0

(%i9) Cournot_Punkt:[XC,PC];
(%o9) $[\, 452\, ,\, 254.0\,]$

b) Man berechne den maximalen Gewinn.
Dazu müssen wir nur mehr die Cournot´sche Menge in die Gewinnfunktion einsetzen.

(%i10) G_max:G(XC);
(%o10) 101252.0

c) Man bestimme die erlösmaximierende Menge und den zugehörigen Preis.
- eine Extremwertaufgabe.

(%i11) U(x);
(%o11) $(480 - 0.5\, x\,)\, x$

(%i12) ab:diff(U(x),x);
(%o12) $480 - 1.0\, x$

(%i13) l:realroots(ab);
(%o13) $[\, x = 480\,]$

(%i14) XU:x,l;
(%o14) 480

```
(%i15) PU:p(XU);
(%o15) 240.0
```

d) Man bestimme den maximalen Erlös.
- die umsatzmaximale Menge in die Umsatzfunktion einsetzen
- Umsatz = Erlös

```
(%i16) U_max:U(XU);
(%o16) 115200.0

(%i17) U_max:PU*XU;
(%o17) 115200.0
```

e) Man bestimme die Gewinngrenzen.
- Grundgleichung: Gewinn = 0
- oder Erlös = Kosten
- untere Gewinngrenze = Gewinnschwelle = Nutzenschwelle = Break Even
- obere Gewinngrenze = Gewinngrenze = Nutzengrenze

(%i18) g1: G(x)=0,expand;

(%o18) $-0.5x^2+452x-900=0$

(%i19) l:realroots(g1),numer;

(%o19) $[x=1.995555549860001, x=902.00444445014]$

```
(%i20) NS:x,l[1];NS:floor(NS*100+0.5)/100.0;
(%o20) 1.995555549860001
(%o21) 2.0

(%i22) NG:x,l[2];NG:floor(NG*100+0.5)/100.0;
(%o22) 902.00444445014
(%o23) 902.0
```

f) Man bestimme die Sättigungsmenge!
- das ist die Nachfrage, wenn der Preis 0 ist

(%i24) p(x);

(%o24) $480-0.5x$

(%i25) l:realroots(p(x));

(%o25) [x = 960]

(%i26) SM:x,l;

(%o26) 960

g) Man bestimme die Preisobergrenze!
- der Preis ist so hoch, dass die Nachfrage 0 wird.

(%i27) p(x);

(%o27) $480 - 0.5\,x$

(%i28) PO:p(0);

(%o28) 480

(%i29) kill(all);

(%o0) *done*

Kompakte Berechnung mit den veränderten Daten: $K=-0{,}005x^2+84x+115200$ und $p=520-0{,}13x$

(%i1) K:K(x):=-0.005*x**2+84*x+115200;

(%o1) $K(x) := (-0.005)\,x^2 + 84\,x + 115200$

(%i2) p:p(x):=520-0.13*x;

(%o2) $p(x) := 520 - 0.13\,x$

(%i3) U(x):=p(x)*x;G(x):=U(x)-K(x);ab:diff(G(x),x);l:realroots(ab);

(%o3) $U(x) := p(x)\,x$

(%o4) $G(x) := U(x) - K(x)$

(%o5) $436 - 0.25\,x$

(%o6) [x = 1744]

(%i7) XC:x,l;PC:p(XC);Cournot_Punkt:[XC,PC];

(%o7) 1744

(%o8) 293.28

(%o9) [1744 , 293.28]

```
(%i10) G_max:G(XC);G_max:floor(G_max*100+0.5)/100.0;
(%o10) 264991.9999999999
(%o11) 264992.0

(%i12) ab:diff(U(x),x);l:realroots(ab);
(%o12) 520 - 0.26 x
(%o13) [ x = 2000 ]

(%i14) XU:x,l;PU:p(XU);U_max:PU*XU;U_max:U(XU);
(%o14) 2000
(%o15) 260.0
(%o16) 520000.0
(%o17) 520000.0

(%i18) l:realroots(G(x)=0);
(%o18) [ x = 288 , x = 3200 ]

(%i19) GS:x,l[1];GG:x,l[2];Gewinnzone:[GS,GG];
(%o19) 288
(%o20) 3200
(%o21) [ 288 , 3200 ]

(%i22) l:realroots(p(x)=0);
(%o22) [ x = 4000 ]

(%i23) XS:x,l;
(%o23) 4000

(%i24) PO:p(0);
(%o24) 520

(%i25) kill(all);
(%o0) done
```

2. Aufgabe (Ein Optimierungsproblem)

Anmerkung: die Aufgabenstellung wird häufig als unrealistisch kritisiert. Eine realistische äquivalente Aufgabenstellung wäre z.B. die Herleitung des Brechungsgesetzes von Snellius aus dem Fermat'schen Prinzip.

Ein Haus liegt 100 m abseits einer geradlinigen Straße, die von einem Fernheizwerk A wegführt. Es soll an das städtische Fernheizsystem angeschlossen werden. Der Laufmeter für die Verlegung kostet längs der Straße € 100,-- je Laufmeter, im Gelände jedoch € 140,--.
An welcher Stelle C muss die Abzweigung erfolgen, damit die Kosten minimal werden? Der hausnächste Punkt B auf der Straße ist 1500 m von A entfernt.
Die Strecke BC soll als Variable x dienen.

Folgende Fragen sind jedenfalls zu lösen:
a) Wie heißt die Kostenfunktion?
b) Für welchen x-Wert hat die Kostenfunktion ein Minimum?
c) Wie groß sind die minimalen Kosten?

Kosten

(%i1) ks:100;kg:140;

(%o1) 100

(%o2) 140

Abmessungen

(%i3) HB:100;AB:1500;BC:x;AC:AB-x;

(%o3) 100

(%o4) 1500

(%o5) x

(%o6) $1500 - x$

Nebenbedingung

(%i7) g:HC**2=HB**2+BC**2 /*es gilt der pythagoräische Lehrsatz */;

(%o7) $HC^2 = x^2 + 10000$

(%i8) l:solve(g,HC);

(%o8) $[\, HC = -\sqrt{x^2 + 10000}\,,\ HC = \sqrt{x^2 + 10000}\,]$

(%i9) HC:ev(HC,l[2]);

(%o9) $\sqrt{x^2 + 10000}$

Kostenfunktion (Lösung von A)

(%i10) K:AC*ks+HC*kg;

(%o10) $140\sqrt{x^2 + 10000} + 100\,(1500 - x)$

Ableitung bestimmen und NULL setzen

(%i11) ab:diff(K,x);

(%o11) $\dfrac{140\,x}{\sqrt{x^2 + 10000}} - 100$

(%i12) g:ab=0;

(%o12) $\dfrac{140\,x}{\sqrt{x^2 + 10000}} - 100 = 0$

Diese Gleichung läßt sich nicht mit Maxima (solve ...) unmittelbar lösen
Man muss "händisch" eingreifen

(%i13) g:g+100;

(%o13) $\dfrac{140\,x}{\sqrt{x^2 + 10000}} = 100$

(%i14) g:g*denom(lhs(g));

(%o14) $140\,x = 100\sqrt{x^2 + 10000}$

(%i15) g:g**2 /* Achtung, das ist keine Äquivalenzumformung */;

(%o15) $19600\,x^2 = 10000\left(x^2 + 10000\right)$

(%i16) l:realroots(g),numer;

(%o16) $[\,x = -102.0620726048946\,,\ x = 102.0620726048946\,]$

Optimale Abzweigung (Lösung von (b)

```
(%i17) X:x,l[2];X:floor(X*100+0.5)/100.0;
(%o17) 102.0620726048946
(%o18) 102.06

(%i19) K_min:K,x=X;K_min:floor(K_min*100+0.5)/100.0;
(%o19) 159797.9589721635
(%o20) 159797.96

(%i21) kill(all);
(%o0) done
```

3. Aufgabe (Finanzmathematik)

a) Jemand nimmt ein Darlehen von 100.000,-- und zahlt jährlich 10.000,-- bei 8% Zinsen dekursiv p.a. zurück. Wann ist er mit den Rückzahlungen fertig?

b) Für einen Kauf gibt es zwei Angebote:
A 100.000,-- sofort, 500.000,-- in 2 Jahren und 400.000,-- in 10 Jahren,
B 300.000,-- sofort und 700.000,-- nach 10 Jahren.
Welches Angebot ist bei einem Kalkulationszinsfuß von i=8% günstiger?

c) Jemand zahlt 20 Jahre lang am Ende jeden Monats 1.000,-- ein. Welche nachschüssige Monatsrente kann er danach 10 Jahre lang beziehen, bei einem Kalkulationszinsfuß von i=8% ?

a) Jemand nimmt ein Darlehen von 100.000,-- und zahlt jährlich 10.000,-- bei 8% Zinsen dekursiv p.a. zurück. Wann ist er mit den Rückzahlungen fertig?

```
(%i1) D:100000;A:10000;p:8;
```

(%o1) 100000

(%o2) 10000

(%o3) 8

```
(%i4) i:p/100;r:1+i,numer;
```

(%o4) $\frac{2}{25}$

(%o5) 1.08

```
(%i6) g:D*r**n=A*(r**n-1)/i,expand;
```

(%o6) $100000\ 1.08^{n} = 125000\ 1.08^{n} - 125000$

Lösung von Exponentialgleichung funktioniert mit solve() schlecht

```
(%i7) g:g-lhs(g);
```

(%o7) $0 = 25000\ 1.08^{n} - 125000$

```
(%i8) g:g+125000;
```

(%o8) $125000 = 25000\ 1.08^{n}$

```
(%i9) g:g/25000;
```
(%o9) $5 = 1.08^{n}$

```
(%i10) g:log(lhs(g))=log(rhs(g));
```
(%o10) $\log(5) = 0.076961041136128\, n$

```
(%i11) l:solve(g,n),numer;
`rat' replaced 1.6094379124341 by 9993/6209 = 1.60943791270736
`rat' replaced -0.07696104113613 by -2495/32419 = -0.07696104136463
`rat' replaced 1.60943791270736 by 8390/5213 = 1.609437943602532
`rat' replaced -0.07696104136463 by -2495/32419 = -0.07696104136463
`rat' replaced -2.958575557584836E-8 by -1/33800049 = -2.958575592597514E-8
`rat' replaced -20.912372221904 by -96176/4599 = -20.912372254838
```
(%o11) $[\, n = 20.91237225483801\,]$

```
(%i12) n:ev(n,l);n:floor(n*100+0.5)/100.0;
(%o12) 20.91237225483801
(%o13) 20.91
```

b) Für einen Kauf gibt es zwei Angebote:
A 100.000,-- sofort, 500.000,-- in 2 Jahren und 400.000,-- in 10 Jahren,
B 300.000,-- sofort und 700.000,-- nach 10 Jahren.
Welches Angebot ist bei einem Kalkulationszinsfuß von i=8% günstiger?

```
(%i14) p:8;i:p/100.0;r:1+i;
(%o14) 8
(%o15) 0.08
(%o16) 1.08

(%i17) A:100000+500000/r**2+400000/r**10;A:floor(A*100+0.5)/100.0;
(%o17) 713946.8053847654
(%o18) 713946.8100000001

(%i19) B:300000+700000/r**10;B:floor(B*100+0.5)/100.0;
(%o19) 624235.4416592789
(%o20) 624235.4399999999
```

```
(%i21) if A>B then "A ist guenstiger"
else if A else "Beide gleich";
(%o21) A ist guenstiger
```

c) Jemand zahlt 20 Jahre lang am Ende jeden Monats 1.000,-- ein. Welche nachschüssige Monatsrente kann er danach 10 Jahre lang beziehen, bei einem Kalkulationszinsfuß von i=8% ?

```
(%i22) R:1000;p:8;n:20;
(%o22) 1000
(%o23) 8
(%o24) 20

(%i25) i:p/100.0;
(%o25) 0.08

(%i26) r:1+i;
(%o26) 1.08

(%i27) rm:r**(1/12);
(%o27) 1.006434030110003

(%i28) im:rm-1;
(%o28) 0.0064340301100034

(%i29) E:R*(rm**(12*n)-1)/im,numer;E:floor(E*100+0.5)/100.0;
(%o29) 568999.0692081694
(%o30) 568999.07

(%i31) B:E;n:10;vm:1/rm;kill(R);
(%o31) 568999.07
(%o32) 10
(%o33) 0.99360710198829
(%o34) done

(%i35) g:B=R*(1-vm**(12*n))/im;
(%o35) 568999.07 = 83.43239038945534 R
```

```
(%i36) l:solve(g,R),numer;
`rat' replaced 568999.07 by 7965987/14 = 568999.0714285715
`rat' replaced -83.4323903894553 by -53063/636 = -83.4323899371069
`rat' replaced 568999.0714285715 by 7396988/13 = 568999.0769230769
`rat' replaced -83.4323899371069 by -53063/636 = -83.4323899371069
`rat' replaced -1.209482341557813E-4 by -1/8268 = -1.209482341557813E-4
`rat' replaced -6819.88227056663 by -115938/17 = -6819.88235294118
(%o36) [ R = 6819.882352941177 ]

(%i37) kill(all);
(%o0) done
```

4. Aufgabe (Wahrscheinlichkeitsrechnung)

a) Belgien hat ungefähr 9,5 Millionen Einwohner, von denen ungefähr ein Drittel Wallonen sind. Wie groß ist die Wahrscheinlichkeit dafür, dass unter 12 zufällig ausgewählten Belgiern
(i) genau 3,
(ii) höchstens 3 Wallonen sind?

b) Ein Kaufgeschäft verkauft durchschnittlich 133 Semmeln pro Tag. Die Streuung beträgt 22. Wie groß ist die Wahrscheinlichkeit einer Verknappung, wenn beim Bäcker täglich 160 Semmeln gekauft werden?

a) Belgien hat ungefähr 9,5 Millionen Einwohner, von denen ungefähr ein Drittel Wallonen sind. Wie groß ist die Wahrscheinlichkeit dafür, dass unter 12 zufällig ausgewählten Belgiern
(i) genau 3,
(ii) höchstens 3 Wallonen sind?

(%i1) n:12;p:1/3;

(%o1) 12

(%o2) $\frac{1}{3}$

(%i3) W(k):=binomial(n,k)*p**k*(1-p)**(n-k);

(%o3) $W(k):=\binom{n}{k} p^{k} (1-p)^{n-k}$

(%i4) WA:W(3),numer;WA:floor(WA*1000+0.5)/1000.0;

(%o4) 0.21195203230462

(%o5) 0.212

(%i6) WB:sum(W(k),k,0,3),numer;WB:floor(WB*1000+0.5)/1000.0;

(%o6) 0.39307467809221

(%o7) 0.393

b) Ein Kaufgeschäft verkauft durchschnittlich 133 Semmeln pro Tag. Die Streuung beträgt 22. Wie groß ist die Wahrscheinlichkeit einer Verknappung, wenn beim Bäcker täglich 160 Semmeln gekauft werden?

(%i8) kill(all);

(%o0) *done*

(%i1) m:133;s:22;x:160;

(%o1) 133

(%o2) 22

(%o3) 160

(%i4) load(distrib);

(%o4)

C:/Programme/Maxima-5.17.1/share/maxima/5.17.1/share/contrib/distrib/distrib.mac

(%i5) W:1-cdf_normal(x,m,s),numer;W:floor(W*1000+0.5)/1000.0;

(%o5) 0.10986005128512

(%o6) 0.11

Created with wxMaxima.